Mechanics - Heavy Symmetric Top 11-16-18

N. T. Gladd

Initialization: Be sure the file *NTGUtilityFunctions.m* is in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[55]:=
```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Background

This is the eighth in a series of Mathematica notebooks on classical mechanics. This series was motivated by a close reading and problem solving project I undertook in 2014. The focus of my attention was the text ***Introduction to Classical Mechanics with Problems and Solutions***, by David Morin. This is a good book from which to learn and has a great collection of problems. I purchased it and recommend that those with interests in this topic acquire it for their library. I do note that an earlier version can be found on the web. This year, when I returned to this project, I decided to focus on generating Mathematica notebooks on material covered in *Chapter 9 Angular Momentum, Part II (General $\vec{L}$),* which deals the 3-D rigid body dynamics. This topic is notorious difficult/confusing for students and I felt I just skimmed by as a graduate student. I return in retirement after all these years to pay my dues and really understand how to solve problems is this area.

Although Morin’s Chapter 9 guides these notebooks, I made frequent use of other sources such as textbooks available in libraries or on the web. I also found lots of video lectures available on YouTube. Confused about some physics topics? Google it and you’ll be amazed what you find. Some relevant texts are

Classical Mechanics, Hebert Goldstein (my original text at University, late 60s). Newer versions exist.
Mechanics: Volume1 A Course in Theoretical Physics, L. D. Landau and E. M. Lifshitz.
Classical Mechanics, John. R. Taylor
Classical Mechanics of Particles and Systems, Stephen T. Thornton, Jerry B. Marion
Analytical Mechanics, G. R. Fowles, G. L. Cassiday

Analytical Mechanics, Louis N. Hand, Janet D. Finch

I find Mathematica useful for this topic. It facilitates calculations, provides a vehicle for creating instructive visualizations and allows one to quickly generate numerical solutions. Mathematica is a favorite tool of mine but I think it is crucially important to also work with pen and paper. Our brains are closely linked to our hands and one thinks differently with a pen in hand than when sitting before a computer screen. For serious thoughts on this, read *The Craftsman*, by Richard Sennett.

Introduction

The following heavy symmetric top is analyzed. “Heavy” here means that the effect of gravity is included. This top is sometimes referred to as the Lagrange top, in contrast with the free symmetric or Euler top.

```
ShowHeavyTop[]
```

Oblate Heavy Top



Figure 1 Heavy top

1 Equations of motion from applied torque perspective

I derive an equation of motion for the symmetric top from

$$\frac{d\vec{L}}{\mathrm{dt}} = \vec{\tau}$$

I first calculate d$\vec{L}$/dt, starting with $\vec{L} = \overleftrightarrow{\mathcal{I}}\cdot\vec{\omega}$.

An Euler angle representation is appropriate (see notebook *Euler's Equations and Euler Angles 11-08-18*). From Section 2 of that notebook the angular frequency can be expressed in terms of body frame unit vectors as

$$\begin{aligned}\vec{\omega} &= \dot{\theta}\, e_1 + \dot{\phi}\, e_z + \dot{\psi}\, e_3 \\ &= \dot{\theta}\, e_1 + \dot{\phi}\,(\sin(\theta)\, e_2 + \cos(\theta)\, e_3) + \dot{\psi}\, e_3 \\ &= \dot{\theta}\, e_1 + \dot{\phi}\sin(\theta)\, e_2 + \left(\dot{\phi}\cos(\theta) + \dot{\psi}\right) e_3 \\ &= \dot{\theta}\, e_1 + \dot{\phi}\sin(\theta)\, e_2 + \dot{\beta}\, e_3\end{aligned}$$

```
w1[1] = 𝕨[t] == θ′[t] 𝕖₁[t] + Sin[θ[t]] ϕ′[t] 𝕖₂[t] + β′[t] 𝕖₃[t]
```

```
𝕨[t] == 𝕖₃[t] β′[t] + 𝕖₁[t] θ′[t] + Sin[θ[t]] 𝕖₂[t] ϕ′[t]
```

I will use double stroke characters to represent vector quantities, but since no double stroke form is available for greek letters I use 𝕨 for $\vec{\omega}$.

```
In[24]:= w["ω equation"] = 𝕨[t] == 𝕖₃[t] β′[t] + 𝕖₁[t] θ′[t] + Sin[θ[t]] 𝕖₂[t] ϕ′[t]
```

```
Out[24]= 𝕨[t] == 𝕖₃[t] β′[t] + 𝕖₁[t] θ′[t] + Sin[θ[t]] 𝕖₂[t] ϕ′[t]
```

I define a variable of convenience

```
In[25]:= def[βDot] = β′[t] == ϕ′[t] Cos[θ[t]] + ψ′[t]
```

```
Out[25]= β′[t] == Cos[θ[t]] ϕ′[t] + ψ′[t]
```

Since $\vec{L} = \overleftrightarrow{\mathcal{I}} \cdot \vec{\omega}$, for a symmetric top with $\mathcal{I}_2 = \mathcal{I}_1$

```
w1[2] = 𝕃[t] == 𝓘₁ 𝕖₁[t] θ′[t] + 𝓘₂ 𝕖₂[t] Sin[θ[t]] ϕ′[t] + 𝕖₃[t] 𝓘₃ β′[t] /. 𝓘₂ → 𝓘₁
```

```
𝕃[t] == 𝓘₃ 𝕖₃[t] β′[t] + 𝓘₁ 𝕖₁[t] θ′[t] + Sin[θ[t]] 𝓘₁ 𝕖₂[t] ϕ′[t]
```

In[26]:=
```
w["L equation"] = 𝕃[t] == 𝓘3 𝕖3[t] β′[t] + 𝓘1 𝕖1[t] θ′[t] + Sin[θ[t]] 𝓘1 𝕖2[t] ϕ′[t]
```

Out[26]=
```
𝕃[t] == 𝓘3 𝕖3[t] β′[t] + 𝓘1 𝕖1[t] θ′[t] + Sin[θ[t]] 𝓘1 𝕖2[t] ϕ′[t]
```

Calculating the time derivative

```
w1[3] = D[w1[2], t]
```

```
𝕃′[t] == Cos[θ[t]] 𝓘1 𝕖2[t] θ′[t] ϕ′[t] + 𝓘1 θ′[t] 𝕖1′[t] + Sin[θ[t]] 𝓘1 ϕ′[t] 𝕖2′[t] +
  𝓘3 β′[t] 𝕖3′[t] + 𝓘3 𝕖3[t] β″[t] + 𝓘1 𝕖1[t] θ″[t] + Sin[θ[t]] 𝓘1 𝕖2[t] ϕ″[t]
```

From notebook Mechanics - *Euler's Equations and Euler Angles 11-08-18*, I have explicit expressions for $\dot{e}_1$, $\dot{e}_2$, $\dot{e}_3$

```
dtRules = {𝕖1′[t] == (Cos[θ[t]] 𝕖2[t] - 𝕖3[t] Sin[θ[t]]) ϕ′[t],
    𝕖2′[t] == 𝕖3[t] θ′[t] - Cos[θ[t]] 𝕖1[t] ϕ′[t],
    𝕖3′[t] == -𝕖2[t] θ′[t] + 𝕖1[t] Sin[θ[t]] ϕ′[t]} // ER;
dtRules // ColumnForm
```

```
𝕖1′[t] → (Cos[θ[t]] 𝕖2[t] - Sin[θ[t]] 𝕖3[t]) ϕ′[t]
𝕖2′[t] → 𝕖3[t] θ′[t] - Cos[θ[t]] 𝕖1[t] ϕ′[t]
𝕖3′[t] → -𝕖2[t] θ′[t] + Sin[θ[t]] 𝕖1[t] ϕ′[t]
```

```
w1[4] = w1[3] /. dtRules
```

```
𝕃′[t] == Cos[θ[t]] 𝓘1 𝕖2[t] θ′[t] ϕ′[t] + 𝓘1 (Cos[θ[t]] 𝕖2[t] - Sin[θ[t]] 𝕖3[t]) θ′[t] ϕ′[t] +
  Sin[θ[t]] 𝓘1 ϕ′[t] (𝕖3[t] θ′[t] - Cos[θ[t]] 𝕖1[t] ϕ′[t]) +
  𝓘3 β′[t] (-𝕖2[t] θ′[t] + Sin[θ[t]] 𝕖1[t] ϕ′[t]) +
  𝓘3 𝕖3[t] β″[t] + 𝓘1 𝕖1[t] θ″[t] + Sin[θ[t]] 𝓘1 𝕖2[t] ϕ″[t]
```

The components are

```
w1[5] = w1[4] /. 𝕃′[t] → L1′[t] 𝕖1[t] + L2′[t] 𝕖2[t] + L3′[t] 𝕖3[t]
```

```
𝕖1[t] L1′[t] + 𝕖2[t] L2′[t] + 𝕖3[t] L3′[t] ==
 Cos[θ[t]] 𝓘1 𝕖2[t] θ′[t] ϕ′[t] + 𝓘1 (Cos[θ[t]] 𝕖2[t] - Sin[θ[t]] 𝕖3[t]) θ′[t] ϕ′[t] +
  Sin[θ[t]] 𝓘1 ϕ′[t] (𝕖3[t] θ′[t] - Cos[θ[t]] 𝕖1[t] ϕ′[t]) +
  𝓘3 β′[t] (-𝕖2[t] θ′[t] + Sin[θ[t]] 𝕖1[t] ϕ′[t]) +
  𝓘3 𝕖3[t] β″[t] + 𝓘1 𝕖1[t] θ″[t] + Sin[θ[t]] 𝓘1 𝕖2[t] ϕ″[t]
```

```
w1[6] = Table[MapEqn[Coefficient[#, arg] &, w1[5] ], {arg, {𝕖1[t], 𝕖2[t], 𝕖3[t]}}]
```

```
{L1′[t] == Sin[θ[t]] 𝓘3 β′[t] ϕ′[t] - Cos[θ[t]] Sin[θ[t]] 𝓘1 ϕ′[t]^2 + 𝓘1 θ″[t],
 L2′[t] == -𝓘3 β′[t] θ′[t] + 2 Cos[θ[t]] 𝓘1 θ′[t] ϕ′[t] + Sin[θ[t]] 𝓘1 ϕ″[t],
 L3′[t] == 𝓘3 β″[t]}
```

Consider the torque produced by gravity at the COM.

$$\vec{\tau} = \vec{r} \times m\,\vec{g} = m\,g\,r\,e_3 \times (-e_z) = m\,g\,r\sin(\theta)\,e_1$$

```
w1[7] = w1[6] /. {L1′[t] → m g r Sin[θ[t]], L2′[t] → 0, L3′[t] → 0};
w1[7] = Reverse /@ w1[7]
```

```
{Sin[θ[t]] I3 β′[t] ϕ′[t] - Cos[θ[t]] Sin[θ[t]] I1 ϕ′[t]^2 + I1 θ′′[t] == g m r Sin[θ[t]],
 -I3 β′[t] θ′[t] + 2 Cos[θ[t]] I1 θ′[t] ϕ′[t] + Sin[θ[t]] I1 ϕ′′[t] == 0, I3 β′′[t] == 0}
```

Finally, the equations of motion for a heavy symmetric top are

```
(TraditionalForm /@ w1[7]) // ColumnForm
```

$$I_3\,\beta'(t)\sin(\theta(t))\,\phi'(t) + I_1\,\theta''(t) + I_1\sin(\theta(t))\,(-\cos(\theta(t)))\,\phi'(t)^2 = g\,m\,r\sin(\theta(t))$$
$$-I_3\,\beta'(t)\,\theta'(t) + 2\,I_1\,\theta'(t)\cos(\theta(t))\,\phi'(t) + I_1\sin(\theta(t))\,\phi''(t) = 0$$
$$I_3\,\beta''(t) = 0$$

The last equation implies $\beta'(t)$ = constant. But since

$$\vec{\omega} = \omega_1\,e_1 + \omega_2\,e_z + \omega_3\,e_3 = \dot{\theta}\,e_1 + \dot{\phi}\sin(\theta)\,e_2 + \dot{\beta}\,e_3$$

the specific value of the constant is ω_{30}

```
w1[8] = w1[7][[1 ;; 2]] /. β′[t] → ω30
```

```
{Sin[θ[t]] I3 ω30 ϕ′[t] - Cos[θ[t]] Sin[θ[t]] I1 ϕ′[t]^2 + I1 θ′′[t] == g m r Sin[θ[t]],
 -I3 ω30 θ′[t] + 2 Cos[θ[t]] I1 θ′[t] ϕ′[t] + Sin[θ[t]] I1 ϕ′′[t] == 0}
```

After a bit of rearrangement

```
w1[9] = w1[8] /. a_ == b_ → a - b == 0
```

```
{-g m r Sin[θ[t]] + Sin[θ[t]] I3 ω30 ϕ′[t] - Cos[θ[t]] Sin[θ[t]] I1 ϕ′[t]^2 + I1 θ′′[t] == 0,
 -I3 ω30 θ′[t] + 2 Cos[θ[t]] I1 θ′[t] ϕ′[t] + Sin[θ[t]] I1 ϕ′′[t] == 0}
```

```
(TraditionalForm /@ w1[9]) // ColumnForm
```

$$-g\,m\,r\sin(\theta(t)) + I_1\,\theta''(t) + \omega_{30}\,I_3\sin(\theta(t))\,\phi'(t) + I_1\sin(\theta(t))\,(-\cos(\theta(t)))\,\phi'(t)^2 = 0$$
$$\omega_{30}\,(-I_3)\,\theta'(t) + 2\,I_1\,\theta'(t)\cos(\theta(t))\,\phi'(t) + I_1\sin(\theta(t))\,\phi''(t) = 0$$

These equations agree with, for example, Morin 9.70

```
In[27]:= w["heavy top equations"] =
  {-g m r Sin[θ[t]] + Sin[θ[t]] I3 ω30 ϕ′[t] - Cos[θ[t]] Sin[θ[t]] I1 ϕ′[t]^2 + I1 θ′′[t] == 0,
   -I3 ω30 θ′[t] + 2 Cos[θ[t]] I1 θ′[t] ϕ′[t] + Sin[θ[t]] I1 ϕ′′[t] == 0}

Out[27]= {-g m r Sin[θ[t]] + Sin[θ[t]] I3 ω30 ϕ′[t] - Cos[θ[t]] Sin[θ[t]] I1 ϕ′[t]^2 + I1 θ′′[t] == 0,
 -I3 ω30 θ′[t] + 2 Cos[θ[t]] I1 θ′[t] ϕ′[t] + Sin[θ[t]] I1 ϕ′′[t] == 0}
```

2 Equations of motion from Lagrangian perspective

I perform an alternative derivation of the equations of motion.

The Lagrangian is

$$\mathcal{L} = \mathcal{T} - \mathcal{V}$$

$$\mathcal{T} = \frac{1}{2} \vec{L} \cdot \vec{\omega}$$

$$\mathcal{V} = m\, g\, r \cos(\theta)$$

Convert equation for $\vec{\omega}$ and $\vec{L}$ to list form

```
w2[1] = w["ω equation"]⟦1⟧ ==
   (Coefficient[w["ω equation"]⟦2⟧, #] & /@ {e₁[t], e₂[t], e₃[t]})
```

```
w[t] == {θ′[t], Sin[θ[t]] ϕ′[t], β′[t]}
```

```
w2[2] = w["L equation"]⟦1⟧ ==
   (Coefficient[w["L equation"]⟦2⟧, #] & /@ {e₁[t], e₂[t], e₃[t]})
```

```
L[t] == {I₁ θ′[t], Sin[θ[t]] I₁ ϕ′[t], I₃ β′[t]}
```

The kinetic energy is

```
def[T] = T == 1/2 Dot[L[t], w[t]]
```

```
T == 1/2 L[t].w[t]
```

I define an alternate, more detailed, definition for the energy

```
(T /. ((def[T] // ER) /. (w2[2] // ER) /. (w2[1] // ER) /.
     β′[t] → ϕ′[t] Cos[θ[t]] + ψ′[t]))
```

$$\frac{1}{2}\left(I_1\, \theta'[t]^2 + \mathrm{Sin}[\theta[t]]^2\, I_1\, \phi'[t]^2 + I_3\,(\mathrm{Cos}[\theta[t]]\, \phi'[t] + \psi'[t])^2\right)$$

In[28]:=
```
def[Tdetail] = T == 1/2 (I₁ θ′[t]^2 + Sin[θ[t]]^2 I₁ ϕ′[t]^2 + I₃ (Cos[θ[t]] ϕ′[t] + ψ′[t])^2)
```

Out[28]=
$$\mathcal{T} == \frac{1}{2}\left(I_1\, \theta'[t]^2 + \mathrm{Sin}[\theta[t]]^2\, I_1\, \phi'[t]^2 + I_3\,(\mathrm{Cos}[\theta[t]]\, \phi'[t] + \psi'[t])^2\right)$$

The potential energy is

In[29]:=
```
def[𝒱] = 𝒱 == m g r Cos[θ[t]]
```
Out[29]=
$\mathcal{V} == g\, m\, r\, \text{Cos}[\theta[t]]$

and the Lagrangian is

```
w2[3] = ℒ == 𝒯 - 𝒱
```
$\mathcal{L} == \mathcal{T} - \mathcal{V}$

```
w2[4] = w2[3] /. (def[𝒯detail] // ER) /. (def[𝒱] // ER)
```
$\mathcal{L} == -g\, m\, r\, \text{Cos}[\theta[t]] + \frac{1}{2}\left(\mathcal{I}_1\, \theta'[t]^2 + \text{Sin}[\theta[t]]^2\, \mathcal{I}_1\, \phi'[t]^2 + \mathcal{I}_3\left(\text{Cos}[\theta[t]]\, \phi'[t] + \psi'[t]\right)^2\right)$

In[30]:=
```
def["Lagrangian"] =
  ℒ == -g m r Cos[θ[t]] + 1/2 (ℐ₁ θ'[t]^2 + Sin[θ[t]]^2 ℐ₁ ϕ'[t]^2 + ℐ₃ (Cos[θ[t]] ϕ'[t] + ψ'[t])^2)
```
Out[30]=
$\mathcal{L} == -g\, m\, r\, \text{Cos}[\theta[t]] + \frac{1}{2}\left(\mathcal{I}_1\, \theta'[t]^2 + \text{Sin}[\theta[t]]^2\, \mathcal{I}_1\, \phi'[t]^2 + \mathcal{I}_3\left(\text{Cos}[\theta[t]]\, \phi'[t] + \psi'[t]\right)^2\right)$

Create a function that facilitates the generation of the Euler Lagrange equations

```
Clear[EulerLagrange];
(* The final t derivative is temporarily suppressed by using 𝒟 instead of D *)
EulerLagrange[ℒ_, dx_, x_] := 𝒟[D[ℒ, dx], t] == D[ℒ, x]
```

Consider the ψ dependence

```
w2[5] = EulerLagrange[def["Lagrangian"][[2]], ψ'[t], ψ[t]]
```
$\mathcal{D}\left[\mathcal{I}_3\left(\text{Cos}[\theta[t]]\, \phi'[t] + \psi'[t]\right), t\right] == 0$

Thus, $\text{Cos}[\theta[t]]\, \phi'[t] + \psi'[t])$ = constant ≡ ω_3

In[31]:=
```
def[ω30] = ω30 == Cos[θ[t]] ϕ'[t] + ψ'[t]
```
Out[31]=
$\omega_{30} == \text{Cos}[\theta[t]]\, \phi'[t] + \psi'[t]$

An alternative form is

In[32]:=
```
def[pψ] = pψ == ℐ₃ (Cos[θ[t]] ϕ'[t] + ψ'[t])
```
Out[32]=
$p_\psi == \mathcal{I}_3\left(\text{Cos}[\theta[t]]\, \phi'[t] + \psi'[t]\right)$

where p_ψ is the canonical momentum corresponding the ψ.

EL equation for ϕ

```
w2[7] = EulerLagrange[def["Lagrangian"][[2]], ϕ′[t], ϕ[t]]
```

$$\mathcal{D}\left[\frac{1}{2}\left(2\,\mathrm{Sin}[\theta[t]]^2\,\mathcal{I}_1\,\phi'[t]+2\,\mathrm{Cos}[\theta[t]]\,\mathcal{I}_3\,\left(\mathrm{Cos}[\theta[t]]\,\phi'[t]+\psi'[t]\right)\right),t\right]==0$$

This can be simplified

```
w2[8] = w2[7] /. Sol[def[ω30], ψ′[t]] // Simplify
```

$$\mathcal{D}\left[\mathrm{Cos}[\theta[t]]\,\mathcal{I}_3\,\omega_{30}+\mathrm{Sin}[\theta[t]]^2\,\mathcal{I}_1\,\phi'[t],t\right]==0$$

Thus, the canonical moment p_ϕ is also constant

In[33]:=
```
def[pϕ] = pϕ == Cos[θ[t]] I3 ω30 + Sin[θ[t]]^2 I1 ϕ′[t]
```

Out[33]=
$$p_\phi==\mathrm{Cos}[\theta[t]]\,\mathcal{I}_3\,\omega_{30}+\mathrm{Sin}[\theta[t]]^2\,\mathcal{I}_1\,\phi'[t]$$

EL equations for θ

```
w2[9] =
 EulerLagrange[def["Lagrangian"][[2]], θ′[t], θ[t]] /. Sol[def[ω30], ψ′[t]] // Simplify
```

$$\mathcal{D}[\mathcal{I}_1\,\theta'[t],t]==\mathrm{Sin}[\theta[t]]\,\left(g\,m\,r-\mathcal{I}_3\,\omega_{30}\,\phi'[t]+\mathrm{Cos}[\theta[t]]\,\mathcal{I}_1\,\phi'[t]^2\right)$$

This is not a constant, so perform the explicit time derivatives

```
w2[10] = {w2[9], w2[8]} /. 𝒟 → D // Simplify
```

$$\{\mathcal{I}_1\,\theta''[t]==\mathrm{Sin}[\theta[t]]\,\left(g\,m\,r-\mathcal{I}_3\,\omega_{30}\,\phi'[t]+\mathrm{Cos}[\theta[t]]\,\mathcal{I}_1\,\phi'[t]^2\right),$$
$$\mathrm{Sin}[\theta[t]]\,\left(-\mathcal{I}_3\,\omega_{30}\,\theta'[t]+\mathcal{I}_1\,\left(2\,\mathrm{Cos}[\theta[t]]\,\theta'[t]\,\phi'[t]+\mathrm{Sin}[\theta[t]]\,\phi''[t]\right)\right)==0\}$$

Standardize the equations

```
w2[11] = w2[10] /. a_ == b_ → a - b == 0 // Reverse
```

$$\{\mathrm{Sin}[\theta[t]]\,\left(-\mathcal{I}_3\,\omega_{30}\,\theta'[t]+\mathcal{I}_1\,\left(2\,\mathrm{Cos}[\theta[t]]\,\theta'[t]\,\phi'[t]+\mathrm{Sin}[\theta[t]]\,\phi''[t]\right)\right)==0,$$
$$-\mathrm{Sin}[\theta[t]]\,\left(g\,m\,r-\mathcal{I}_3\,\omega_{30}\,\phi'[t]+\mathrm{Cos}[\theta[t]]\,\mathcal{I}_1\,\phi'[t]^2\right)+\mathcal{I}_1\,\theta''[t]==0\}$$

Factor the common term in the second equation.

```
w2[12] = {MapEqn[(#/Sin[θ[t]]) &, w2[11][[1]]], w2[11][[2]]} // Expand // Reverse
```

$$\{-g\,m\,r\,\mathrm{Sin}[\theta[t]]+\mathrm{Sin}[\theta[t]]\,\mathcal{I}_3\,\omega_{30}\,\phi'[t]-\mathrm{Cos}[\theta[t]]\,\mathrm{Sin}[\theta[t]]\,\mathcal{I}_1\,\phi'[t]^2+\mathcal{I}_1\,\theta''[t]==0,$$
$$-\mathcal{I}_3\,\omega_{30}\,\theta'[t]+2\,\mathrm{Cos}[\theta[t]]\,\mathcal{I}_1\,\theta'[t]\,\phi'[t]+\mathrm{Sin}[\theta[t]]\,\mathcal{I}_1\,\phi''[t]==0\}$$

This result should agree with the calculation of $d\vec{L}/dt = \vec{\tau}$ in the previous section.

In[34]:=
```
w["heavy top equations"]
```

Out[34]=
$$\{-g\, m\, r\, \mathrm{Sin}[\theta[t]] + \mathrm{Sin}[\theta[t]]\, \mathcal{I}_3\, \omega_{30}\, \phi'[t] - \mathrm{Cos}[\theta[t]]\, \mathrm{Sin}[\theta[t]]\, \mathcal{I}_1\, \phi'[t]^2 + \mathcal{I}_1\, \theta''[t] == 0,\ -\mathcal{I}_3\, \omega_{30}\, \theta'[t] + 2\,\mathrm{Cos}[\theta[t]]\, \mathcal{I}_1\, \theta'[t]\, \phi'[t] + \mathrm{Sin}[\theta[t]]\, \mathcal{I}_1\, \phi''[t] == 0\}$$

```
{w2[12][[1, 1]] - w["heavy top equations"][[1, 1]],
  w2[12][[2, 1]] - w["heavy top equations"][[2, 1]]} // Simplify
```

{0, 0}

3 Precession of top in the special case θ = constant

Consider the heavy top equations for the special case $\theta = \theta_0$ constant

```
w3[1] = w["heavy top equations"]
```

$$\{-g\, m\, r\, \mathrm{Sin}[\theta[t]] + \mathrm{Sin}[\theta[t]]\, \mathcal{I}_3\, \omega_{30}\, \phi'[t] - \mathrm{Cos}[\theta[t]]\, \mathrm{Sin}[\theta[t]]\, \mathcal{I}_1\, \phi'[t]^2 + \mathcal{I}_1\, \theta''[t] == 0,\ -\mathcal{I}_3\, \omega_{30}\, \theta'[t] + 2\,\mathrm{Cos}[\theta[t]]\, \mathcal{I}_1\, \theta'[t]\, \phi'[t] + \mathrm{Sin}[\theta[t]]\, \mathcal{I}_1\, \phi''[t] == 0\}$$

```
w3[2] = w3[1] /. θ → Function[{t}, θ0]
```

$$\{-g\, m\, r\, \mathrm{Sin}[\theta_0] + \mathrm{Sin}[\theta_0]\, \mathcal{I}_3\, \omega_{30}\, \phi'[t] - \mathrm{Cos}[\theta_0]\, \mathrm{Sin}[\theta_0]\, \mathcal{I}_1\, \phi'[t]^2 == 0,\ \mathrm{Sin}[\theta_0]\, \mathcal{I}_1\, \phi''[t] == 0\}$$

The second equation implies $\phi'[t]$ = constant ≡ ω_ϕ

```
w3[3] = w3[2][[1]] /. ϕ'[t] → ωϕ
```

$$-g\, m\, r\, \mathrm{Sin}[\theta_0] + \mathrm{Sin}[\theta_0]\, \mathcal{I}_3\, \omega_{30}\, \omega_\phi - \mathrm{Cos}[\theta_0]\, \mathrm{Sin}[\theta_0]\, \mathcal{I}_1\, \omega_\phi^2 == 0$$

This frequency has two possible values

```
w3[4] = Solve[w3[3], ωϕ] // RE
```

$$\left\{\left\{\omega_\phi == -\frac{1}{2\,\mathcal{I}_1}\mathrm{Sec}[\theta_0]\left(-\mathcal{I}_3\,\omega_{30} - \sqrt{-4\, g\, m\, r\, \mathrm{Cos}[\theta_0]\, \mathcal{I}_1 + \mathcal{I}_3^2\,\omega_{30}^2}\right)\right\},\ \left\{\omega_\phi == -\frac{1}{2\,\mathcal{I}_1}\mathrm{Sec}[\theta_0]\left(-\mathcal{I}_3\,\omega_{30} + \sqrt{-4\, g\, m\, r\, \mathrm{Cos}[\theta_0]\, \mathcal{I}_1 + \mathcal{I}_3^2\,\omega_{30}^2}\right)\right\}\right\}$$

The physical interpretation of quantities in the numerator are

$\mathcal{I}_3\,\omega_{30}$ angular momentum along the spin axis L_3

g m r Cos[θ_0] angular momentum due to gravitational torque

Consider the high frequency limit in which the spin angular momentum dominates the applied gravitational torque. To facilitate the calculation, introduce

```
def[η] = η == (4 g m r Cos[θ0] I1)/(I3^2 ω30^2)
```

$$\eta == \frac{4\,g\,m\,r\,\mathrm{Cos}[\theta_0]\,\mathcal{I}_1}{\mathcal{I}_3^2\,\omega_{30}^2}$$

```
w3[5] = w3[4] /. Solve[def[η], g][[1]]
```

$$\left\{\left\{\omega_\phi == -\frac{1}{2\,\mathcal{I}_1}\mathrm{Sec}[\theta_0]\left(-\mathcal{I}_3\,\omega_{30}-\sqrt{\mathcal{I}_3^2\,\omega_{30}^2-\eta\,\mathcal{I}_3^2\,\omega_{30}^2}\right)\right\},\left\{\omega_\phi == -\frac{1}{2\,\mathcal{I}_1}\mathrm{Sec}[\theta_0]\left(-\mathcal{I}_3\,\omega_{30}+\sqrt{\mathcal{I}_3^2\,\omega_{30}^2-\eta\,\mathcal{I}_3^2\,\omega_{30}^2}\right)\right\}\right\}$$

```
w3[6] = MapEqn[Simplify[#, {I3 > 0, ω3 > 0}] &, w3[5]]
```

$$\left\{\left\{\omega_\phi == \frac{1}{2\,\mathcal{I}_1}\mathrm{Sec}[\theta_0]\,\mathcal{I}_3\left(\omega_{30}+\sqrt{-(-1+\eta)\,\omega_{30}^2}\right)\right\},\left\{\omega_\phi == -\frac{1}{2\,\mathcal{I}_1}\mathrm{Sec}[\theta_0]\,\mathcal{I}_3\left(-\omega_{30}+\sqrt{-(-1+\eta)\,\omega_{30}^2}\right)\right\}\right\}$$

```
w3[7] = MapEqn[Normal@Series[#, {η, 0, 1}] &, w3[6]];
w3[7] = Simplify[w3[7], {ω30 > 0}]
```

$$\left\{\left\{\omega_\phi == -\frac{(-4+\eta)\,\mathrm{Sec}[\theta_0]\,\mathcal{I}_3\,\omega_{30}}{4\,\mathcal{I}_1}\right\},\left\{\frac{\eta\,\mathrm{Sec}[\theta_0]\,\mathcal{I}_3\,\omega_{30}}{\mathcal{I}_1} == 4\,\omega_\phi\right\}\right\}$$

```
w3[7] = MapEqn[Normal@Series[#, {η, 0, 1}] &, w3[6]];
w3[7] = MapEqn[Simplify[#, {ω30 > 0}] &, w3[7]]
```

$$\left\{\left\{\omega_\phi == -\frac{(-4+\eta)\,\mathrm{Sec}[\theta_0]\,\mathcal{I}_3\,\omega_{30}}{4\,\mathcal{I}_1}\right\},\left\{\omega_\phi == \frac{\eta\,\mathrm{Sec}[\theta_0]\,\mathcal{I}_3\,\omega_{30}}{4\,\mathcal{I}_1}\right\}\right\}$$

Reverting to the original variables

```
w3[8] = w3[7] /. (def[η] // ER)
```

$$\left\{\left\{\omega_\phi == -\frac{1}{4\,\mathcal{I}_1}\mathrm{Sec}[\theta_0]\,\mathcal{I}_3\left(-4+\frac{4\,g\,m\,r\,\mathrm{Cos}[\theta_0]\,\mathcal{I}_1}{\mathcal{I}_3^2\,\omega_{30}^2}\right)\omega_{30}\right\},\left\{\omega_\phi == \frac{g\,m\,r}{\mathcal{I}_3\,\omega_{30}}\right\}\right\}$$

The fast precession (first term) is essentially the precession of a free symmetric top (slightly slowed by the effect of the gravitational torque. The slow precession (second term) is solely due to the gravitational torque.

Note also from w3[4] that there would be no precession if $\omega_3 < \omega_{3,\mathrm{min}}$ where

```
w3[9] = (Solve[-4 g m r Cos[θ] 𝓘₁ + 𝓘₃² ω₃² == 0, ω₃]⟦2, 1⟧ // RE) /. ω₃ -> ω3min
```

$$\omega_{3\,\mathrm{min}} == \frac{2 \sqrt{g} \sqrt{m} \sqrt{r} \sqrt{\mathrm{Cos}[\theta]} \sqrt{\mathcal{I}_1}}{\mathcal{I}_3}$$

4 Single equation of motion

The two heavy top equations for θ and ϕ can be reduced to a single equation in θ. This is instructive since the precessional motion of the top has an analogy to the familiar problems of the motion of a mass in a potential well.

The energy of the top is

```
w4[1] = ℰ == 𝒯 + 𝒱
```

$$\mathcal{E} == \mathcal{T} + \mathcal{V}$$

In detail

```
w4[2] = w4[1] /. (def[𝒯detail] // ER) /. (def[𝒱] // ER)
```

$$\mathcal{E} == g\,m\,r\,\mathrm{Cos}[\theta[t]] + \frac{1}{2}\left(\mathcal{I}_1\,\theta'[t]^2 + \mathrm{Sin}[\theta[t]]^2\,\mathcal{I}_1\,\phi'[t]^2 + \mathcal{I}_3\left(\mathrm{Cos}[\theta[t]]\,\phi'[t] + \psi'[t]\right)^2\right)$$

For future reference

In[35]:=
```
def[ℰ] =
  ℰ == g m r Cos[θ[t]] + 1/2 (𝓘₁ θ'[t]^2 + Sin[θ[t]]^2 𝓘₁ ϕ'[t]^2 + 𝓘₃ (Cos[θ[t]] ϕ'[t] + ψ'[t])^2)
```

Out[35]=
$$\mathcal{E} == g\,m\,r\,\mathrm{Cos}[\theta[t]] + \frac{1}{2}\left(\mathcal{I}_1\,\theta'[t]^2 + \mathrm{Sin}[\theta[t]]^2\,\mathcal{I}_1\,\phi'[t]^2 + \mathcal{I}_3\left(\mathrm{Cos}[\theta[t]]\,\phi'[t] + \psi'[t]\right)^2\right)$$

Introduce the dimensionless time parameter T = ω_{30}t. I argue that the natural frequency for the top is the spin frequency ω_{30}.

In[36]:=
```
def[T] = T == ω30 t
```

Out[36]=
$$T == t\,\omega_{30}$$

```
w4[3] = w4[2] /. {θ → Function[{t}, θ[ω30 t]],
    ϕ → Function[{t}, ϕ[ω30 t]], ψ → Function[{t}, ψ[ω30 t]]} /. Sol[def[T], t]
```

$$\mathcal{E} == g\,m\,r\,\mathrm{Cos}[\theta[T]] + \frac{1}{2}\left(\mathcal{I}_1\,\omega_{30}^2\,\theta'[T]^2 + \mathrm{Sin}[\theta[T]]^2\,\mathcal{I}_1\,\omega_{30}^2\,\phi'[T]^2 + \mathcal{I}_3\left(\mathrm{Cos}[\theta[T]]\,\omega_{30}\,\phi'[T] + \omega_{30}\,\psi'[T]\right)^2\right)$$

Normalize with respect to spin kinetic energy 1/2 $I_3\ \omega^2{}_{30}$

```
w4[4] = MapEqn[(#/(I3 ω30^2/2)) &, w4[3]] // Expand
```

$$\frac{2\,\mathcal{E}}{\mathcal{I}_3\,\omega_{30}^2} == \frac{2\,g\,m\,r\,\mathrm{Cos}[\theta[T]]}{\mathcal{I}_3\,\omega_{30}^2} + \frac{\mathcal{I}_1\,\theta'[T]^2}{\mathcal{I}_3} + \mathrm{Cos}[\theta[T]]^2\,\phi'[T]^2 + \frac{\mathrm{Sin}[\theta[T]]^2\,\mathcal{I}_1\,\phi'[T]^2}{\mathcal{I}_3} + 2\,\mathrm{Cos}[\theta[T]]\,\phi'[T]\,\psi'[T] + \psi'[T]^2$$

Introduce some dimensionless parameters

In[37]:=
```
def[𝔼] = 𝔼 == 2 ℰ / (I3 ω30^2)
```

Out[37]=
$$\mathbb{E} == \frac{2\,\mathcal{E}}{\mathcal{I}_3\,\omega_{30}^2}$$

In[38]:=
```
def[𝔾] = 𝔾 == 2 g m r / (I3 ω30^2)
```

Out[38]=
$$\mathbb{G} == \frac{2\,g\,m\,r}{\mathcal{I}_3\,\omega_{30}^2}$$

and a parameter for the geometry of the top

In[39]:=
```
def[κ] = κ == I3 / I1
```

Out[39]=
$$\kappa == \frac{\mathcal{I}_3}{\mathcal{I}_1}$$

```
w4[5] = w4[4] /. Sol[def[𝔼], ℰ] /. Sol[def[𝔾], g] /. Sol[def[κ], I3]
```

$$\mathbb{E} == \mathbb{G}\,\mathrm{Cos}[\theta[T]] + \frac{\theta'[T]^2}{\kappa} + \mathrm{Cos}[\theta[T]]^2\,\phi'[T]^2 + \frac{\mathrm{Sin}[\theta[T]]^2\,\phi'[T]^2}{\kappa} + 2\,\mathrm{Cos}[\theta[T]]\,\phi'[T]\,\psi'[T] + \psi'[T]^2$$

```
w4[6] = Solve[w4[5] /. θ′[T]^2 → x, x]⟦1, 1⟧ /. x → θ′[T]^2
```

$\theta'[T]^2 \to \mathbb{E}\,\kappa - \mathbb{G}\,\kappa\,\text{Cos}[\theta[T]] - \kappa\,\text{Cos}[\theta[T]]^2\,\phi'[T]^2 - \text{Sin}[\theta[T]]^2\,\phi'[T]^2 - 2\,\kappa\,\text{Cos}[\theta[T]]\,\phi'[T]\,\psi'[T] - \kappa\,\psi'[T]^2$

The constants of motion of this problem are the energy and the canonical momenta p_ϕ and p_ψ introduced in Section 2. These latter two can be used to remove the ϕ and ψ dependence of the last equation.

Consider p_ψ

```
w4[7] = def[pψ]
```

$p_\psi == \mathcal{I}_3\,(\text{Cos}[\theta[t]]\,\phi'[t] + \psi'[t])$

```
w4[8] =
 w4[7]⟦1⟧ == (w4[7]⟦2⟧ /. {θ → Function[{t}, θ[ω30 t]], ϕ → Function[{t}, ϕ[ω30 t]],
      ψ → Function[{t}, ψ[ω30 t]]} /. Sol[def[T], t] )
```

$p_\psi == \mathcal{I}_3\,(\text{Cos}[\theta[T]]\,\omega_{30}\,\phi'[T] + \omega_{30}\,\psi'[T])$

```
w4[9] = MapEqn[(Simplify[# / (I3 ω30)]) &, w4[8]]
```

$\frac{p_\psi}{\mathcal{I}_3\,\omega_{30}} == \text{Cos}[\theta[T]]\,\phi'[T] + \psi'[T]$

In[40]:=
```
def[ℙψ] = ℙψ == pψ/(I3 ω30)
```

Out[40]=
$\mathbb{P}_\psi == \frac{p_\psi}{\mathcal{I}_3\,\omega_{30}}$

In[41]:=
```
w4[10] = w4[9] /. Sol[def[ℙψ], pψ]
```

Out[41]=
w4[9]

Similarly

```
w4[11] = def[pϕ]
```

$p_\phi == \text{Cos}[\theta[t]]\,\mathcal{I}_3\,\omega_{30} + \text{Sin}[\theta[t]]^2\,\mathcal{I}_1\,\phi'[t]$

```
w4[12] = w4[11][[1]] ==
    (w4[11][[2]] /. {θ → Function[{t}, θ[ω30 t]], ϕ → Function[{t}, ϕ[ω30 t]],
        ψ → Function[{t}, ψ[ω30 t]]} /. Sol[def[T], t]);
w4[12] = MapEqn[(Simplify[#/(I3 ω30)]) &, w4[12]]
```

$$\frac{p_\phi}{\mathcal{I}_3\,\omega_{30}} == \mathrm{Cos}[\theta[T]] + \frac{\mathrm{Sin}[\theta[T]]^2\,\mathcal{I}_1\,\phi'[T]}{\mathcal{I}_3}$$

In[42]:=

```
def[ℙϕ] = ℙϕ == pϕ/(I3 ω30)
```

Out[42]=

$$\mathbb{P}_\phi == \frac{p_\phi}{\mathcal{I}_3\,\omega_{30}}$$

```
w4[13] = w4[12] /. Sol[def[ℙϕ], pϕ] /. Sol[def[κ], I3]
```

$$\mathbb{P}_\phi == \mathrm{Cos}[\theta[T]] + \frac{\mathrm{Sin}[\theta[T]]^2\,\phi'[T]}{\kappa}$$

The constant momenta equations yield θ dependent expressions for $\phi'[T]$ and $\psi'[T]$

```
w4[14] = Solve[{w4[10], w4[13]}, {ϕ'[T], ψ'[T]}][[1]]
```

$$\{\phi'[T] \to -\kappa\,\mathrm{Cot}[\theta[T]]\,\mathrm{Csc}[\theta[T]] + \kappa\,\mathrm{Csc}[\theta[T]]^2\,\mathbb{P}_\phi,$$
$$\psi'[T] \to \kappa\,\mathrm{Cot}[\theta[T]]^2 - \kappa\,\mathrm{Cot}[\theta[T]]\,\mathrm{Csc}[\theta[T]]\,\mathbb{P}_\phi + \mathbb{P}_\psi\}$$

Using these results in the equation for $\theta'[T]$ the following single ode is obtained

```
w4[15] = w4[6] /. w4[14] // Simplify // RE
```

$$\theta'[T]^2 ==$$
$$-\kappa\,(-\mathbb{E} + \mathbb{G}\,\mathrm{Cos}[\theta[T]] + \kappa\,\mathrm{Cot}[\theta[T]]^2 - 2\,\kappa\,\mathrm{Cot}[\theta[T]]\,\mathrm{Csc}[\theta[T]]\,\mathbb{P}_\phi + \kappa\,\mathrm{Csc}[\theta[T]]^2\,\mathbb{P}_\phi^2 + \mathbb{P}_\psi^2)$$

Isolate the θ dependence on the RHS

```
w4[16] = Select[w4[15][[2, 3]], (Not[FreeQ[#, θ]]) &]
```

$$\mathbb{G}\,\mathrm{Cos}[\theta[T]] + \kappa\,\mathrm{Cot}[\theta[T]]^2 - 2\,\kappa\,\mathrm{Cot}[\theta[T]]\,\mathrm{Csc}[\theta[T]]\,\mathbb{P}_\phi + \kappa\,\mathrm{Csc}[\theta[T]]^2\,\mathbb{P}_\phi^2$$

and identify it as a potential

```
def[V] = V == 𝔾 Cos[θ[T]] + κ Cot[θ[T]]^2 - 2 κ Cot[θ[T]] Csc[θ[T]] ℙϕ + κ Csc[θ[T]]^2 ℙϕ^2
```

$$V == \mathbb{G}\,\mathrm{Cos}[\theta[T]] + \kappa\,\mathrm{Cot}[\theta[T]]^2 - 2\,\kappa\,\mathrm{Cot}[\theta[T]]\,\mathrm{Csc}[\theta[T]]\,\mathbb{P}_\phi + \kappa\,\mathrm{Csc}[\theta[T]]^2\,\mathbb{P}_\phi^2$$

```
w4[16] = w4[15] /. Sol[def[V], ℙϕ];
w4[16] = MapEqn[Simplify, w4[16]]
```

$$\theta'[T]^2 == -\kappa\left(V - \mathbb{E} + \mathbb{P}_\psi^2\right)$$

Equations of this form are familiar from, say, the motion of a body under a central force.

4-1 Numerical example

I will define a function for V. First I make some preparatory changes since it is awkward to use sub-scripts within a Mathematica function

```
𝔾 Cos[θ[T]] + κ Cot[θ[T]]^2 - 2 κ Cot[θ[T]] Csc[θ[T]] ℙϕ + κ Csc[θ[T]]^2 ℙϕ^2 /.
 {θ[T] → θ, ℙϕ → ℙϕ}
```

$$\mathbb{G}\,\mathrm{Cos}[\theta] + \kappa\,\mathrm{Cot}[\theta]^2 - 2\,\mathbb{P}\phi\,\kappa\,\mathrm{Cot}[\theta]\,\mathrm{Csc}[\theta] + \mathbb{P}\phi^2\,\kappa\,\mathrm{Csc}[\theta]^2$$

```
Clear[Potential];
Potential[θ_, ℙϕ_, 𝔾_, κ_] :=
 𝔾 Cos[θ] + κ Cot[θ]^2 - 2 ℙϕ κ Cot[θ] Csc[θ] + ℙϕ^2 κ Csc[θ]^2
```

I plot this potential for some nominal parameters. For the oblate ellipsoidal top in Figure 1, we have a = 1, b = 1/2. In notebook Mechanics - *Free Symmetric Top 11-10-18*, I calculated the inertia tensor

$$\begin{pmatrix} \frac{1}{5}\left(a^2+b^2\right)M & 0 & 0 \\ 0 & \frac{1}{5}\left(a^2+b^2\right)M & 0 \\ 0 & 0 & \frac{2a^2M}{5} \end{pmatrix}$$

Specific values for the relevant dimensionless parameters are needed to visualize *Potential*

```
w41[1] = {I1 → (a^2 + b^2) M / 5, I3 → (2 a^2) M / 5} /. {a → 1, b → 1/2, M → 1}
```

$$\left\{\mathcal{I}_1 \to \frac{1}{4}, \mathcal{I}_3 \to \frac{2}{5}\right\}$$

Thus

```
w41[2] = def[κ] /. w41[1]
```

$$\kappa == \frac{8}{5}$$

Next, I need values for $\mathbb{P}\phi$ and $\mathbb{P}\psi$. Recall

```
{def[ℙ_ϕ], def[p_ϕ]}
```

$$\left\{\mathbb{P}_\phi == \frac{p_\phi}{\mathcal{I}_3\,\omega_{30}},\ p_\phi == \mathrm{Cos}[\theta[t]]\,\mathcal{I}_3\,\omega_{30} + \mathrm{Sin}[\theta[t]]^2\,\mathcal{I}_1\,\phi'[t]\right\}$$

Again, consistently with Figure 1, I will arbitrarily choose the initial orientation of the top to be θ(t = 0) = π/6, as well as take θ'(0) = 0.10 ω_{30}. I will assume this is a rapidly spinning top so ψ'(0) = ω_{30} >> 1. I will assume the precession to be much slower, say ϕ'(0) = 0.05 ω_{30}

```
w41[3] = def[p_ϕ] /. t → 0 /. θ[0] → π / 6 /. ϕ'[0] → ω30/20 /. Sol[def[κ], I1]
```

$$p_\phi == \frac{1}{2}\sqrt{3}\,\mathcal{I}_3\,\omega_{30} + \frac{\mathcal{I}_3\,\omega_{30}}{80\,\kappa}$$

for which

```
w41[4] = def[ℙ_ϕ] /. (w41[3] // ER);
w41[4] = MapEqn[Simplify, w41[4]]
```

$$\mathbb{P}_\phi == \frac{1}{80}\left(40\sqrt{3} + \frac{1}{\kappa}\right)$$

Next, a value for the parameter $\mathbb{P}_\phi$

```
w41[5] = w41[4] /. κ → 8/5 // N
```

$$\mathbb{P}_\phi == 0.873838$$

Recall

```
{def[ℙ_ψ], def[p_ψ]}
```

$$\left\{\mathbb{P}_\psi == \frac{p_\psi}{\mathcal{I}_3\,\omega_{30}},\ p_\psi == \mathcal{I}_3\left(\mathrm{Cos}[\theta[t]]\,\phi'[t] + \psi'[t]\right)\right\}$$

```
w41[6] = def[p_ψ] /. t → 0 /. θ[0] → π / 6 /. ϕ'[0] → ω30/20 /. ψ'[0] → ω30 /.
  Sol[def[κ], I1]
```

$$p_\psi == \mathcal{I}_3\left(\omega_{30} + \frac{\sqrt{3}\,\omega_{30}}{40}\right)$$

```
w41[7] = def[ℙ_ψ] /. (w41[6] // ER);
w41[7] = MapEqn[Simplify, w41[7]]
```

$$\mathbb{P}_\psi == 1 + \frac{\sqrt{3}}{40}$$

```
ℙψ == 1 + √3/40 // N
```

$\mathbb{P}_\psi == 1.0433$

Recall

```
def[𝔾]
```

$$\mathbb{G} == \frac{2\,g\,m\,r}{\mathcal{I}_3\,\omega_{30}^2}$$

I will assume the initial gravitational potential energy is small with respect to the spin kinetic energy and choose

```
w41[8] = 𝔾 == 1/10
```

$$\mathbb{G} == \frac{1}{10}$$

For the energy parameter

```
def[𝔼]
```

$$\mathbb{E} == \frac{2\,\mathcal{E}}{\mathcal{I}_3\,\omega_{30}^2}$$

```
w41[9] = def[𝔼] /. Sol[def[ℰ], ℰ]  // Expand
```

$$\mathbb{E} == \frac{2\,g\,m\,r\,\mathrm{Cos}[\theta[t]]}{\mathcal{I}_3\,\omega_{30}^2} + \frac{\mathcal{I}_1\,\theta'[t]^2}{\mathcal{I}_3\,\omega_{30}^2} + \frac{\mathrm{Cos}[\theta[t]]^2\,\phi'[t]^2}{\omega_{30}^2} + \frac{\mathrm{Sin}[\theta[t]]^2\,\mathcal{I}_1\,\phi'[t]^2}{\mathcal{I}_3\,\omega_{30}^2} + \frac{2\,\mathrm{Cos}[\theta[t]]\,\phi'[t]\,\psi'[t]}{\omega_{30}^2} + \frac{\psi'[t]^2}{\omega_{30}^2}$$

```
w41[10] = w41[9] /. t → 0 /. Sol[def[𝔾], g] /. Sol[def[κ], 𝓘₃]
```

$$\mathbb{E} == \mathbb{G}\,\mathrm{Cos}[\theta[0]] + \frac{\theta'[0]^2}{\kappa\,\omega_{30}^2} + \frac{\mathrm{Cos}[\theta[0]]^2\,\phi'[0]^2}{\omega_{30}^2} + \frac{\mathrm{Sin}[\theta[0]]^2\,\phi'[0]^2}{\kappa\,\omega_{30}^2} + \frac{2\,\mathrm{Cos}[\theta[0]]\,\phi'[0]\,\psi'[0]}{\omega_{30}^2} + \frac{\psi'[0]^2}{\omega_{30}^2}$$

```
w41[11] = w41[10] /.
  {θ[0] → π / 6, θ′[0] → 0.1 ω30, ϕ[0] → 0, ϕ′[0] → 0.05 ω30, ψ[0] → 0, ψ′[0] → ω30}
```

$$\mathbb{E} == 1.08848 + \frac{\sqrt{3}\,\mathbb{G}}{2} + \frac{0.010625}{\kappa}$$

Recall the equation of motion

$$\theta'[T]^2 == -\kappa\left(V - \mathbb{E} + \mathbb{P}_\psi^2\right)$$

```
Module[{ℙϕ = 0.87383, ℙψ = 1.0433, 𝔾 = 0.1,
  κ = 8/5., 𝔼, TPinner, TPouter, refLine, points, lab},
 𝔼 = 1.08848 + (√3 𝔾)/2 + 0.010625/κ;
 TPinner = FindRoot[Potential[θ, ℙϕ, 𝔾, κ] == 𝔼 - ℙψ^2, {θ, π/8}][[1, 2]];
 TPouter = FindRoot[Potential[θ, ℙϕ, 𝔾, κ] == 𝔼 - ℙψ^2, {θ, π / 4}][[1, 2]];
 refLine = {Red, Line[{{0, 𝔼 - ℙψ^2}, {π, 𝔼 - ℙψ^2}}] };
 points = {Red, PointSize[0.015], Point[{TPinner, Potential[TPinner, ℙϕ, 𝔾, κ]}],
   Point[{TPouter, Potential[TPouter, ℙϕ, 𝔾, κ]}]};
 lab = Stl@StringForm["Potential well for θ motion\nℙϕ = `` ℙψ = ``, 𝔾 = ``,
      κ = ``, 𝔼 = ``\ninner TP = ``π, outer TP = ``π", NF2@ℙϕ, NF2@ℙψ,
    NF2@𝔾, NF2@κ, NF2@𝔼, NF2[N[TPinner/π]], NF2[N[TPouter/π]]];
 Plot[Potential[θ, ℙϕ, 𝔾, κ], {θ, 0, π/2}, AxesLabel → {Stl["θ"], Stl["V[θ]"]},
  PlotStyle → Black, PlotRange → {Automatic, 0.5},
  Ticks → {{0, π/8, π / 4, 3 π/8, π/2, 5 π/8, 3 π / 4}, Automatic},
  PlotLabel → lab, Epilog → {refLine, points}]]
```



Figure 2 Potential well for θ motion of the heavy top

5 Numerical solution of heavy top equations

In this section I generate a numerical solution of the original heavy top equations

```
w5[1] = w["heavy top equations"]
```

$$\{-g\,m\,r\,\mathrm{Sin}[\theta[t]] + \mathrm{Sin}[\theta[t]]\,\mathcal{I}_3\,\omega_{30}\,\phi'[t] - \mathrm{Cos}[\theta[t]]\,\mathrm{Sin}[\theta[t]]\,\mathcal{I}_1\,\phi'[t]^2 + \mathcal{I}_1\,\theta''[t] == 0,\ -\mathcal{I}_3\,\omega_{30}\,\theta'[t] + 2\,\mathrm{Cos}[\theta[t]]\,\mathcal{I}_1\,\theta'[t]\,\phi'[t] + \mathrm{Sin}[\theta[t]]\,\mathcal{I}_1\,\phi''[t] == 0\}$$

I derive a dimensionless form, using the same method as in the previous Section.

```
w5[2] = w5[1] /. {θ → Function[{t}, θ[ω30 t]],
    ϕ → Function[{t}, ϕ[ω30 t]], ψ → Function[{t}, ψ[ω30 t]]} /. Sol[def[T], t]
```

$$\{-g\,m\,r\,\mathrm{Sin}[\theta[T]] + \mathrm{Sin}[\theta[T]]\,\mathcal{I}_3\,\omega_{30}^2\,\phi'[T] - \mathrm{Cos}[\theta[T]]\,\mathrm{Sin}[\theta[T]]\,\mathcal{I}_1\,\omega_{30}^2\,\phi'[T]^2 + \mathcal{I}_1\,\omega_{30}^2\,\theta''[T] == 0,\ -\mathcal{I}_3\,\omega_{30}^2\,\theta'[T] + 2\,\mathrm{Cos}[\theta[T]]\,\mathcal{I}_1\,\omega_{30}^2\,\theta'[T]\,\phi'[T] + \mathrm{Sin}[\theta[T]]\,\mathcal{I}_1\,\omega_{30}^2\,\phi''[T] == 0\}$$

```
w5[3] = MapEqn[(Expand[#/(I3 ω30^2/2)]) &, w5[2]]
```

$$\left\{-\frac{2\,g\,m\,r\,\mathrm{Sin}[\theta[T]]}{\mathcal{I}_3\,\omega_{30}^2} + 2\,\mathrm{Sin}[\theta[T]]\,\phi'[T] - \frac{2\,\mathrm{Cos}[\theta[T]]\,\mathrm{Sin}[\theta[T]]\,\mathcal{I}_1\,\phi'[T]^2}{\mathcal{I}_3} + \frac{2\,\mathcal{I}_1\,\theta''[T]}{\mathcal{I}_3} == 0,\ -2\,\theta'[T] + \frac{4\,\mathrm{Cos}[\theta[T]]\,\mathcal{I}_1\,\theta'[T]\,\phi'[T]}{\mathcal{I}_3} + \frac{2\,\mathrm{Sin}[\theta[T]]\,\mathcal{I}_1\,\phi''[T]}{\mathcal{I}_3} == 0\right\}$$

```
w5[4] = w5[3] /. Sol[def[𝔾], g] /. Sol[def[κ], I1]
```

$$\left\{-\mathbb{G}\,\mathrm{Sin}[\theta[T]] + 2\,\mathrm{Sin}[\theta[T]]\,\phi'[T] - \frac{2\,\mathrm{Cos}[\theta[T]]\,\mathrm{Sin}[\theta[T]]\,\phi'[T]^2}{\kappa} + \frac{2\,\theta''[T]}{\kappa} == 0,\ -2\,\theta'[T] + \frac{4\,\mathrm{Cos}[\theta[T]]\,\theta'[T]\,\phi'[T]}{\kappa} + \frac{2\,\mathrm{Sin}[\theta[T]]\,\phi''[T]}{\kappa} == 0\right\}$$

```
w5[5] = MapEqn[(Expand[# κ/2]) &, w5[4]]
```

$$\left\{-\frac{1}{2}\,\mathbb{G}\,\kappa\,\mathrm{Sin}[\theta[T]] + \kappa\,\mathrm{Sin}[\theta[T]]\,\phi'[T] - \mathrm{Cos}[\theta[T]]\,\mathrm{Sin}[\theta[T]]\,\phi'[T]^2 + \theta''[T] == 0,\ -\kappa\,\theta'[T] + 2\,\mathrm{Cos}[\theta[T]]\,\theta'[T]\,\phi'[T] + \mathrm{Sin}[\theta[T]]\,\phi''[T] == 0\right\}$$

In[43]:=

```
def["Heavy top eqns - dimensionless"] =
  {-1/2 𝔾 κ Sin[θ[T]] + κ Sin[θ[T]] ϕ'[T] - Cos[θ[T]] Sin[θ[T]] ϕ'[T]^2 + θ''[T] == 0,
   -κ θ'[T] + 2 Cos[θ[T]] θ'[T] ϕ'[T] + Sin[θ[T]] ϕ''[T] == 0};
```

These need to be combined with initial conditions

```
w5[6] = Join[w5[5], {ϕ[0] == ϕ0, ϕ′[0] == dϕ0, θ[0] == θ0, θ′[0] == dθ0 }]
```

$$\left\{-\frac{1}{2}\,\mathbb{G}\,\kappa\,\mathrm{Sin}[\theta[T]] + \kappa\,\mathrm{Sin}[\theta[T]]\,\phi'[T] - \mathrm{Cos}[\theta[T]]\,\mathrm{Sin}[\theta[T]]\,\phi'[T]^2 + \theta''[T] == 0,\right.$$
$$-\kappa\,\theta'[T] + 2\,\mathrm{Cos}[\theta[T]]\,\theta'[T]\,\phi'[T] + \mathrm{Sin}[\theta[T]]\,\phi''[T] == 0,$$
$$\left.\phi[0] == \phi0,\ \phi'[0] == d\phi0,\ \theta[0] == \theta0,\ \theta'[0] == d\theta0\right\}$$

```
(TraditionalForm /@ w5[6]) // ColumnForm
```

$$-\frac{1}{2}\,\mathbb{G}\,\kappa\,\sin(\theta(T)) + \theta''(T) + \kappa\,\sin(\theta(T))\,\phi'(T) + \sin(\theta(T))\,(-\cos(\theta(T)))\,\phi'(T)^2 = 0$$
$$-\kappa\,\theta'(T) + 2\,\theta'(T)\,\cos(\theta(T))\,\phi'(T) + \sin(\theta(T))\,\phi''(T) = 0$$
$$\phi(0) = \phi0$$
$$\phi'(0) = d\phi0$$
$$\theta(0) = \theta0$$
$$\theta'(0) = d\theta0$$

I solve these numerically for the same parameters used in Section 4.1.

In[59]:=
```
TMAX = 50;
SOLNODES =
 Module[{θ0 = π / 6, dθ0 = 0.1, ϕ0 = 0, dϕ0 = 0.05, κ = 8/5., 𝔾 = 0.1, eqns, soln},
  eqns = {-1/2 𝔾 κ Sin[θ[T]] + κ Sin[θ[T]] ϕ′[T] - Cos[θ[T]] Sin[θ[T]] ϕ′[T]^2 + θ″[T] == 0,
    -κ θ′[T] + 2 Cos[θ[T]] θ′[T] ϕ′[T] + Sin[θ[T]] ϕ″[T] == 0,
    ϕ[0] == ϕ0, ϕ′[0] == dϕ0, θ[0] == θ0, θ′[0] == dθ0};
  soln = NDSolve[eqns, {ϕ, θ}, {T, 0, TMAX}][[1]] ]
```

Out[60]=
{ϕ → InterpolatingFunction[ Domain: {{0., 50.}} Output: scalar],

θ → InterpolatingFunction[ Domain: {{0., 50.}} Output: scalar]}

```
Plot[{ϕ[T] /. SOLNODES, θ[T] /. SOLNODES}, {T, 0, TMAX}, PlotStyle → {Black, Blue},
 AxesLabel → {Stl["T"], ""}, PlotLegends → {Stl["ϕ"], Stl["θ"]}]
```



This result is better visualized in the context of the body frame of the top. The motion of the top involves both precession and nutation. The red lines represent the turning points in θ as calculated in Figure 2 of Section 4.1.

```
Module[{scale = 1.25, δ = 0.1, viewPoint = {2.4, 1, 1}, 𝒦, eqns,
  soln, O, ex, ey, ez, axes, StoC, turningPoints, GBackground, G},
 StoC[r_, θ_, ϕ_] := {r Cos[ϕ] Sin[θ], r Sin[θ] Sin[ϕ], r Cos[θ]};

 {O, ex, ey, ez} = {{0, 0, 0}, {1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
 axes = {Black, Line[scale {-ex, ex}], Line[scale {-ey, ey}],
   Line[scale {- ez, ez}], Stl@Text["x", (scale + δ) ex],
   Stl@ Text["y", (scale + δ) ey], Stl@Text["z", (scale + δ) ez]};
 turningPoints = {Red, Line@Table[StoC[1, 0.15 π, α], {α, 0, 2 π, 2 π / 36}],
   Line@Table[StoC[1, 0.19 π, α], {α, 0, 2 π, 2 π / 36}]};
 GBackground = Graphics3D[{{Opacity[.25], Sphere[]}, turningPoints, axes},
   ViewPoint → viewPoint, Boxed → False];

 G = ParametricPlot3D[
   (StoC[1, θ[T], ϕ[T]] /. SOLNODES), {T, 0, TMAX}, PlotStyle → Black];
 Show[GBackground, G]]
```



Figure 3 Representative precession and nutation of a heavy top

In the lab frame, the precessing and nutating top wobbles.

```
Module[{T = 0, a = 1, b = 0.5, r = 1, θ = -π / 6, frames},
 frames = Table[ShowWobblingTopUsingEulerMatrix[T, r, a, b], {T, 0, 25, 1}];
 ListAnimate[frames]]
```



In[44]:=
```
Clear[ShowWobblingTopUsingEulerMatrix];
ShowWobblingTopUsingEulerMatrix[T_, r_, a_, b_] :=
 Module[{scale = 2.5, δ = 0.1, vp = {2.5, 1.0, 1}, sz = 0.01,
   szArrow = 0.03, O, ex, ey, ez, labAxes, e1, e2, e3, (* ℐ1,
   ℐ3,ω1, ω2, ω3, ωVec,LVec, *), base, top, ellipsoid, bodyAxes,
   objects, EM, range, latlongcurves, RM = RotationMatrix, lab, G},

  range = scale {{- 1, 1}, {- 1, 1}, {- 1, 1}};
  {O, ex, ey, ez} = {{0, 0, 0}, {1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
  labAxes = {Black, Line[1.3 {-ex, ex}], Line[1.3 { -ey, ey}],
    Line[1.3 {-ez, ez}], Tex["x", 1.5 ex], Tex["y", 1.5 ey], Tex["z", 1.9 ez ] };
  {e1, e2, e3} = {{1, 0, 0}, {0, 1, 0}, {0, 0, 1}};

  latlongcurves =
   {LightGray, Table[Line@Table[{a Sin[θθ] Cos[ϕϕ], a Sin[θθ] Sin[ϕϕ], b Cos[θθ] },
       {θθ, 0, π, π/100}], {ϕϕ, 0, 2 π, π/12}],
    Table[Line@Table[{a Sin[θθ] Cos[ϕϕ], a Sin[θθ] Sin[ϕϕ], b Cos[θθ] },
       {ϕϕ, 0, 2 π, π/100}], {θθ, -π, π, π/8}]};
  bodyAxes = {Blue, Tube[1.3 a {-e1, e1}], Tube[1.3 a { -e2, e2}], Tube[1.3 b {-e3, e3}],
    Tex["x₁", 1.5 a e1], Tex["x₂", 1.5 a e2], Tex["x₃", 1.5 b e3] };
  base = {LightGreen, Polygon[2 {{-1, -1, 0}, {-1, 1, 0}, {1, 1, 0}, {1, -1, 0}}]};
  top = {{Opacity[0.25, Blue], Ellipsoid[O, {a, a, b}]}, latlongcurves,
    {Black, bodyAxes}, {Blue, Tube[{{0, 0, r}, {0, 0, -r}}]}};
  top = GeometricTransformation[top, TranslationTransform[{0, 0, r}]];

  (* The top has been constructed in the body frame. It is transformed into the body
   frame using the Euler matrix and the numerically determined values of ϕ and ψ *)
  EM = EulerMatrix[{ϕ[T], θ[T], T}, {3, 1, 3}] /. SOLNODES;
  top = GeometricTransformation[top, EM];

  lab = Module[{type },
    type = Which[a == b, "Spheroid",
      a < b, "Prolate",
      a > b, "Oblate"];
    Stl@StringForm["Wobbling `` top in lab frame\n a = ``, b = ``, r = ``",
      type, NF2@a, NF2@b, NF2@r]];
  Graphics3D[{labAxes, base, top}, ImageSize → 400, Axes → False, Boxed → False,
   SphericalRegion → True, ViewPoint → vp, PlotLabel → lab, PlotRange → range]]
```

6 Morin's approximate model for describing precession and nutation

There are many discussions of the precession and nutation of a heavy top in textbooks and the literature. I think the treatment of Morin in Section 9.7.7 is particularly illuminating. I reproduce that calculation in this section.

Start with the dimensionless form of the heavy top equations

```
w6[1] = def["Heavy top eqns - dimensionless"]
```

$$\left\{-\frac{1}{2}\mathbb{G}\,\kappa\,\mathrm{Sin}[\theta[T]]+\kappa\,\mathrm{Sin}[\theta[T]]\,\phi'[T]-\mathrm{Cos}[\theta[T]]\,\mathrm{Sin}[\theta[T]]\,\phi'[T]^2+\theta''[T]==0,\right.$$
$$\left.-\kappa\,\theta'[T]+2\,\mathrm{Cos}[\theta[T]]\,\theta'[T]\,\phi'[T]+\mathrm{Sin}[\theta[T]]\,\phi''[T]==0\right\}$$

Perturb these equations, using $\phi \sim \epsilon\,\phi$, $\theta \sim \epsilon\,\theta$ where $\epsilon << 1$. Also, assume that $\theta(T) \simeq \theta_m$, where θ_m is near the bottom of the potential well V.

```
w6[2] =
 w6[1] /. {ϕ → Function[{T}, ϵ ϕ[T]], θ → Function[{T}, ϵ θ[T]]} /. θ[T] → θm
```

$$\left\{-\frac{1}{2}\mathbb{G}\,\kappa\,\mathrm{Sin}[\epsilon\,\theta m]+\epsilon\,\kappa\,\mathrm{Sin}[\epsilon\,\theta m]\,\phi'[T]-\epsilon^2\,\mathrm{Cos}[\epsilon\,\theta m]\,\mathrm{Sin}[\epsilon\,\theta m]\,\phi'[T]^2+\epsilon\,\theta''[T]==0,\right.$$
$$\left.-\epsilon\,\kappa\,\theta'[T]+2\,\epsilon^2\,\mathrm{Cos}[\epsilon\,\theta m]\,\theta'[T]\,\phi'[T]+\epsilon\,\mathrm{Sin}[\epsilon\,\theta m]\,\phi''[T]==0\right\}$$

The linear terms are

```
w6[3] = w6[2] /. ϵ^2 → 0 /. ϵ → 1
```

$$\left\{-\frac{1}{2}\mathbb{G}\,\kappa\,\mathrm{Sin}[\theta m]+\kappa\,\mathrm{Sin}[\theta m]\,\phi'[T]+\theta''[T]==0,\ -\kappa\,\theta'[T]+\mathrm{Sin}[\theta m]\,\phi''[T]==0\right\}$$

Differentiate the second equation and solve for $\theta''[T]$

```
w6[4] = Solve[w6[3][[2]], θ'[T]][[1, 1]] // RE;
w6[4] = MapEqn[(D[#, T]) &, w6[4]]
```

$$\theta''[T]==\frac{\mathrm{Sin}[\theta m]\,\phi^{(3)}[T]}{\kappa}$$

Use this in the first equation

```
w6[5] = w6[3][[1]] /. (w6[4] // ER)
```

$$-\frac{1}{2}\mathbb{G}\,\kappa\,\mathrm{Sin}[\theta m]+\kappa\,\mathrm{Sin}[\theta m]\,\phi'[T]+\frac{\mathrm{Sin}[\theta m]\,\phi^{(3)}[T]}{\kappa}==0$$

Write this in a form that facilitates solving for $\phi'[T]$

```
w6[6] = w6[5] /. ϕ'[T] → ξ[T] /. ϕ^(3)[T] → ξ''[T]
```

$$-\frac{1}{2}\mathbb{G}\,\kappa\,\mathrm{Sin}[\theta m]+\kappa\,\mathrm{Sin}[\theta m]\,\xi[T]+\frac{\mathrm{Sin}[\theta m]\,\xi''[T]}{\kappa}==0$$

```
w6[7] = DSolve[w6[6], ξ[T], T][[1, 1]] /. ξ[T] → ϕ′[T] // RE
```

$$\phi'[T] == \frac{\mathbb{G}}{2} + C[1]\,\text{Cos}[T\,\kappa] + C[2]\,\text{Sin}[T\,\kappa]$$

To be consistent with Morin’s development I rewrite the oscillatory terms

```
w6[8] = w6[7] /. C[1] Cos[T κ] + C[2] Sin[T κ] → 𝒜 Cos[κ T + ℬ]
```

$$\phi'[T] == \frac{\mathbb{G}}{2} + \mathcal{A}\,\text{Cos}[\mathcal{B} + T\,\kappa]$$

where $\mathcal{A}$ and $\mathcal{B}$ are constants of integration.

Integrate this result

```
w6[9] = ϕ[T] == Simplify@Integrate[w6[8][[2]], T] + 𝒞
```

$$\phi[T] == \mathcal{C} + \frac{T\,\mathbb{G}}{2} + \frac{\mathcal{A}\,\text{Sin}[\mathcal{B} + T\,\kappa]}{\kappa}$$

where $\mathcal{C}$ is a constant of integration.

To obtain an expression for θ, recall w6[3]

```
w6[10] = w6[3][[2]]
```

$$-\kappa\,\theta'[T] + \text{Sin}[\theta m]\,\phi''[T] == 0$$

Using the expression for ϕ

```
w6[11] = w6[10] /. ϕ → Function[{T}, T 𝔾/2 + 𝒜 Sin[ℬ + T κ]/κ]
```

$$-\mathcal{A}\,\kappa\,\text{Sin}[\theta m]\,\text{Sin}[\mathcal{B} + T\,\kappa] - \kappa\,\theta'[T] == 0$$

```
w6[12] = Solve[w6[11], θ′[T]][[1, 1]] // RE
```

$$\theta'[T] == -\mathcal{A}\,\text{Sin}[\theta m]\,\text{Sin}[\mathcal{B} + T\,\kappa]$$

Integrate

```
w6[13] = θ[T] == Simplify@Integrate[w6[12][[2]], T] + 𝒟
```

$$\theta[T] == \mathcal{D} + \frac{\mathcal{A}\,\text{Cos}[\mathcal{B} + T\,\kappa]\,\text{Sin}[\theta m]}{\kappa}$$

where $\mathcal{D}$ is a constant of integration.

Collect the results

```
w6[14] = {w6[13], w6[9]}
```

$$\left\{\Theta[T] == \mathcal{D} + \frac{\mathcal{A}\,\text{Cos}[\mathcal{B} + T\,\kappa]\,\text{Sin}[\Theta m]}{\kappa},\ \phi[T] == C + \frac{T\,\mathbb{G}}{2} + \frac{\mathcal{A}\,\text{Sin}[\mathcal{B} + T\,\kappa]}{\kappa}\right\}$$

and the derivatives

```
w6[15] = MapEqn[D[#, T] &, w6[14]]
```

$$\left\{\Theta'[T] == -\mathcal{A}\,\text{Sin}[\Theta m]\,\text{Sin}[\mathcal{B} + T\,\kappa],\ \phi'[T] == \frac{\mathbb{G}}{2} + \mathcal{A}\,\text{Cos}[\mathcal{B} + T\,\kappa]\right\}$$

In[46]:=

$$\text{def["Morin equations"]} = \left\{\Theta[T] == \mathcal{D} + \frac{\mathcal{A}\,\text{Cos}[\mathcal{B} + T\,\kappa]\,\text{Sin}[\Theta m]}{\kappa},\ \phi[T] == C + \frac{T\,\mathbb{G}}{2} + \frac{\mathcal{A}\,\text{Sin}[\mathcal{B} + T\,\kappa]}{\kappa},\ \Theta'[T] == -\mathcal{A}\,\text{Sin}[\Theta m]\,\text{Sin}[\mathcal{B} + T\,\kappa],\ \phi'[T] == \frac{\mathbb{G}}{2} + \mathcal{A}\,\text{Cos}[\mathcal{B} + T\,\kappa]\right\};$$

```
(TraditionalForm /@ def["Morin equations"]) // ColumnForm
```

$$\Theta(T) = \mathcal{D} + \frac{\mathcal{A}\sin(\Theta m)\cos(\kappa T + \mathcal{B})}{\kappa}$$
$$\phi(T) = C + \frac{\mathcal{A}\sin(\kappa T + \mathcal{B})}{\kappa} + \frac{\mathbb{G}\,T}{2}$$
$$\Theta'(T) = -\mathcal{A}\sin(\Theta m)\sin(\kappa T + \mathcal{B})$$
$$\phi'(T) = \frac{\mathbb{G}}{2} + \mathcal{A}\cos(\kappa T + \mathcal{B})$$

These are consistent with Morin (9.88) and (9.90).

6.1 Numerical analysis of the Morin results

I want to compare θ(T) and ϕ(T) for the Morin approximate treatment with the numerical solution obtained in Section 5. I need specific values for the initial conditions

```
w61[1] = def["Morin equations"]
```

$$\left\{\Theta[T] == \mathcal{D} + \frac{\mathcal{A}\,\text{Cos}[\mathcal{B} + T\,\kappa]\,\text{Sin}[\Theta m]}{\kappa},\ \phi[T] == C + \frac{T\,\mathbb{G}}{2} + \frac{\mathcal{A}\,\text{Sin}[\mathcal{B} + T\,\kappa]}{\kappa},\ \Theta'[T] == -\mathcal{A}\,\text{Sin}[\Theta m]\,\text{Sin}[\mathcal{B} + T\,\kappa],\ \phi'[T] == \frac{\mathbb{G}}{2} + \mathcal{A}\,\text{Cos}[\mathcal{B} + T\,\kappa]\right\}$$

```
w61[2] = w61[1] /. T → 0 /. {Θ[0] → Θ0, ϕ[0] → ϕ0, Θ'[0] → dΘ0, ϕ'[0] → dϕ0}
```

$$\left\{\Theta 0 == \mathcal{D} + \frac{\mathcal{A}\,\text{Cos}[\mathcal{B}]\,\text{Sin}[\Theta m]}{\kappa},\ \phi 0 == C + \frac{\mathcal{A}\,\text{Sin}[\mathcal{B}]}{\kappa},\ d\Theta 0 == -\mathcal{A}\,\text{Sin}[\mathcal{B}]\,\text{Sin}[\Theta m],\ d\phi 0 == \frac{\mathbb{G}}{2} + \mathcal{A}\,\text{Cos}[\mathcal{B}]\right\}$$

```
w61[3] = Solve[w61[2], {𝒜, ℬ, 𝒞, 𝒟}]
```

$$\left\{\left\{\mathcal{A}\to-\frac{1}{2}\,\mathrm{Csc}[\theta m]\,\sqrt{\left(4\,d\theta0^2+4\,d\phi0^2\,\mathrm{Sin}[\theta m]^2-4\,d\phi0\,\mathbb{G}\,\mathrm{Sin}[\theta m]^2+\mathbb{G}^2\,\mathrm{Sin}[\theta m]^2\right)},\right.\right.$$
$$\mathcal{C}\to\frac{\mathrm{Csc}[\theta m]\,\left(d\theta0+\kappa\,\phi0\,\mathrm{Sin}[\theta m]\right)}{\kappa},$$
$$\mathcal{D}\to\frac{1}{2\,\kappa}\left(2\,\theta0\,\kappa-2\,d\phi0\,\mathrm{Sin}[\theta m]+\mathbb{G}\,\mathrm{Sin}[\theta m]\right),\ \mathcal{B}\to\mathrm{ConditionalExpression}\Big[$$
$$\mathrm{ArcTan}\Big[\Big(-2\,d\phi0\,\mathrm{Sin}[\theta m]\,\sqrt{\left(4\,d\theta0^2+4\,d\phi0^2\,\mathrm{Sin}[\theta m]^2-4\,d\phi0\,\mathbb{G}\,\mathrm{Sin}[\theta m]^2+\mathbb{G}^2\,\mathrm{Sin}[\theta m]^2\right)}+$$
$$\mathbb{G}\,\mathrm{Sin}[\theta m]\,\sqrt{\left(4\,d\theta0^2+4\,d\phi0^2\,\mathrm{Sin}[\theta m]^2-4\,d\phi0\,\mathbb{G}\,\mathrm{Sin}[\theta m]^2+\mathbb{G}^2\,\mathrm{Sin}[\theta m]^2\right)}\Big)\Big/$$
$$\left(4\,d\theta0^2+4\,d\phi0^2\,\mathrm{Sin}[\theta m]^2-4\,d\phi0\,\mathbb{G}\,\mathrm{Sin}[\theta m]^2+\mathbb{G}^2\,\mathrm{Sin}[\theta m]^2\right),$$
$$\left(2\,d\theta0\right)\Big/\left(\sqrt{\left(4\,d\theta0^2+4\,d\phi0^2\,\mathrm{Sin}[\theta m]^2-4\,d\phi0\,\mathbb{G}\,\mathrm{Sin}[\theta m]^2+\mathbb{G}^2\,\mathrm{Sin}[\theta m]^2\right)}\right)\Big]+$$
$$\left.2\,\pi\,C[1],\ C[1]\in\mathrm{Integers}\Big]\right\},$$
$$\left\{\mathcal{A}\to\frac{1}{2}\,\mathrm{Csc}[\theta m]\,\sqrt{\left(4\,d\theta0^2+4\,d\phi0^2\,\mathrm{Sin}[\theta m]^2-4\,d\phi0\,\mathbb{G}\,\mathrm{Sin}[\theta m]^2+\mathbb{G}^2\,\mathrm{Sin}[\theta m]^2\right)},\right.$$
$$\mathcal{C}\to\frac{\mathrm{Csc}[\theta m]\,\left(d\theta0+\kappa\,\phi0\,\mathrm{Sin}[\theta m]\right)}{\kappa},$$
$$\mathcal{D}\to\frac{1}{2\,\kappa}\left(2\,\theta0\,\kappa-2\,d\phi0\,\mathrm{Sin}[\theta m]+\mathbb{G}\,\mathrm{Sin}[\theta m]\right),$$
$$\mathcal{B}\to\mathrm{ConditionalExpression}\Big[$$
$$\mathrm{ArcTan}\Big[\Big(-2\,d\phi0\,\mathrm{Sin}[\theta m]\,\sqrt{\left(4\,d\theta0^2+4\,d\phi0^2\,\mathrm{Sin}[\theta m]^2-4\,d\phi0\,\mathbb{G}\,\mathrm{Sin}[\theta m]^2+\mathbb{G}^2\,\mathrm{Sin}[\theta m]^2\right)}+$$
$$\mathbb{G}\,\mathrm{Sin}[\theta m]\,\sqrt{\left(4\,d\theta0^2+4\,d\phi0^2\,\mathrm{Sin}[\theta m]^2-4\,d\phi0\,\mathbb{G}\,\mathrm{Sin}[\theta m]^2+\mathbb{G}^2\,\mathrm{Sin}[\theta m]^2\right)}\Big)\Big/$$
$$\left(-4\,d\theta0^2-4\,d\phi0^2\,\mathrm{Sin}[\theta m]^2+4\,d\phi0\,\mathbb{G}\,\mathrm{Sin}[\theta m]^2-\mathbb{G}^2\,\mathrm{Sin}[\theta m]^2\right),$$
$$-\left(\left(2\,d\theta0\right)\Big/\left(\sqrt{\left(4\,d\theta0^2+4\,d\phi0^2\,\mathrm{Sin}[\theta m]^2-4\,d\phi0\,\mathbb{G}\,\mathrm{Sin}[\theta m]^2+\mathbb{G}^2\,\mathrm{Sin}[\theta m]^2\right)}\right)\right)\Big]+$$
$$\left.\left.2\,\pi\,C[1],\ C[1]\in\mathrm{Integers}\Big]\right\}\right\}$$

Choose the “phase” parameter $\mathcal{B}$ to lie between 0 and 2 π

```
w61[4] = w61[3] /. C[1] → 0 // Simplify
```

$$\left\{\left\{\mathcal{A}\to-\frac{1}{2}\,\mathrm{Csc}[\theta m]\,\sqrt{\left(\left(\left(-2\,d\phi0+\mathbb{G}\right)^2+4\,d\theta0^2\,\mathrm{Csc}[\theta m]^2\right)\mathrm{Sin}[\theta m]^2\right)},\right.\right.$$
$$\mathcal{C}\to\phi0+\frac{d\theta0\,\mathrm{Csc}[\theta m]}{\kappa},\ \mathcal{D}\to\frac{2\,\theta0\,\kappa+\left(-2\,d\phi0+\mathbb{G}\right)\,\mathrm{Sin}[\theta m]}{2\,\kappa},$$
$$\mathcal{B}\to\mathrm{ArcTan}\Big[\left(\left(-2\,d\phi0+\mathbb{G}\right)\,\mathrm{Sin}[\theta m]\right)\Big/\left(\sqrt{\left(4\,d\theta0^2+\left(-2\,d\phi0+\mathbb{G}\right)^2\,\mathrm{Sin}[\theta m]^2\right)}\right),$$
$$\left.\left(2\,d\theta0\right)\Big/\left(\sqrt{\left(4\,d\theta0^2+\left(-2\,d\phi0+\mathbb{G}\right)^2\,\mathrm{Sin}[\theta m]^2\right)}\right)\Big]\right\},$$
$$\left\{\mathcal{A}\to\frac{1}{2}\,\mathrm{Csc}[\theta m]\,\sqrt{\left(\left(\left(-2\,d\phi0+\mathbb{G}\right)^2+4\,d\theta0^2\,\mathrm{Csc}[\theta m]^2\right)\mathrm{Sin}[\theta m]^2\right)},\right.$$
$$\mathcal{C}\to\phi0+\frac{d\theta0\,\mathrm{Csc}[\theta m]}{\kappa},\ \mathcal{D}\to\frac{2\,\theta0\,\kappa+\left(-2\,d\phi0+\mathbb{G}\right)\,\mathrm{Sin}[\theta m]}{2\,\kappa},$$
$$\mathcal{B}\to\mathrm{ArcTan}\Big[\left(\left(2\,d\phi0-\mathbb{G}\right)\,\mathrm{Sin}[\theta m]\right)\Big/\left(\sqrt{\left(4\,d\theta0^2+\left(-2\,d\phi0+\mathbb{G}\right)^2\,\mathrm{Sin}[\theta m]^2\right)}\right),$$
$$\left.\left.-\left(\left(2\,d\theta0\right)\Big/\left(\sqrt{\left(4\,d\theta0^2+\left(-2\,d\phi0+\mathbb{G}\right)^2\,\mathrm{Sin}[\theta m]^2\right)}\right)\right)\Big]\right\}\right\}$$

Choose the branch with positive values of the constants of integration

```
w61[5] = w61[4][[2]] // RE
```

$$\left\{\mathcal{A} == \frac{1}{2}\,\mathrm{Csc}[\theta m]\,\sqrt{\left(\left((-2\,d\phi 0 + \mathbb{G})^2 + 4\,d\theta 0^2\,\mathrm{Csc}[\theta m]^2\right)\mathrm{Sin}[\theta m]^2\right)},\right.$$
$$\mathcal{C} == \phi 0 + \frac{d\theta 0\,\mathrm{Csc}[\theta m]}{\kappa},\ \mathcal{D} == \frac{2\,\theta 0\,\kappa + (-2\,d\phi 0 + \mathbb{G})\,\mathrm{Sin}[\theta m]}{2\,\kappa},\ \mathcal{B} == \mathrm{ArcTan}\Big[$$
$$\left.\frac{(2\,d\phi 0 - \mathbb{G})\,\mathrm{Sin}[\theta m]}{\sqrt{4\,d\theta 0^2 + (-2\,d\phi 0 + \mathbb{G})^2\,\mathrm{Sin}[\theta m]^2}},\ -\left((2\,d\theta 0)\Big/\left(\sqrt{\left(4\,d\theta 0^2 + (-2\,d\phi 0 + \mathbb{G})^2\,\mathrm{Sin}[\theta m]^2\right)}\right)\right)\Big]\right\}$$

Using the same initial conditions as for the numerical solution in Section 5, the constants of integration are

$$\mathrm{Module}\Big[\{\theta m = \pi/6,\ \kappa = 8/5,\ \mathbb{G} = 0.1,\ \theta 0 = \pi/6,\ d\theta 0 = 0.1,\ \phi 0 = 0,\ d\phi 0 = 0.05\},$$
$$\left\{\mathcal{A} == \frac{1}{2}\,\mathrm{Csc}[\theta m]\,\sqrt{\left(\left((-2\,d\phi 0 + \mathbb{G})^2 + 4\,d\theta 0^2\,\mathrm{Csc}[\theta m]^2\right)\mathrm{Sin}[\theta m]^2\right)},\right.$$
$$\mathcal{C} == \phi 0 + \frac{d\theta 0\,\mathrm{Csc}[\theta m]}{\kappa},\ \mathcal{D} == \frac{2\,\theta 0\,\kappa + (-2\,d\phi 0 + \mathbb{G})\,\mathrm{Sin}[\theta m]}{2\,\kappa},$$
$$\mathcal{B} == \mathrm{ArcTan}\Big[\left((2\,d\phi 0 - \mathbb{G})\,\mathrm{Sin}[\theta m]\right)\Big/\left(\sqrt{\left(4\,d\theta 0^2 + (-2\,d\phi 0 + \mathbb{G})^2\,\mathrm{Sin}[\theta m]^2\right)}\right),$$
$$\left.-\left((2\,d\theta 0)\Big/\left(\sqrt{\left(4\,d\theta 0^2 + (-2\,d\phi 0 + \mathbb{G})^2\,\mathrm{Sin}[\theta m]^2\right)}\right)\right)\Big]\right\}\Big]$$

$$\{\mathcal{A} == 0.2,\ \mathcal{C} == 0.125,\ \mathcal{D} == 0.523599,\ \mathcal{B} == -1.5708\}$$

I compare Morin’s approximate solution with the numerical solution obtained in Section 5

```
Module[{θm = π / 6,  κ = 8 / 5,  𝔾 = 0.1,
  θ0 = π / 6, dθ0 = 0.1, ϕ0 = 0, dϕ0 = 0.05, 𝒜, ℬ, 𝒞, 𝒟, lab},
 {𝒜 = 1/2 Csc[θm] √(((-2 dϕ0 + 𝔾)^2 + 4 dθ0^2 Csc[θm]^2) Sin[θm]^2),
  𝒞 = ϕ0 + (dθ0 Csc[θm])/κ, 𝒟 = (2 θ0 κ + (-2 dϕ0 + 𝔾) Sin[θm])/(2 κ),
  ℬ = ArcTan[((2 dϕ0 - 𝔾) Sin[θm]) / (√(4 dθ0^2 + (-2 dϕ0 + 𝔾)^2 Sin[θm]^2)),
    -((2 dθ0) / (√(4 dθ0^2 + (-2 dϕ0 + 𝔾)^2 Sin[θm]^2)))]};
 lab = Stl@StringForm["θm = ``, κ = ``, 𝔾 = ``\nθ0 = ``, dθ0 = ``,
      ϕ0 = ``, dϕ0 = ``\n𝒜 = ``, ℬ = ``, 𝒞 = ``, 𝒟 = ``", θm, κ,
    NF2@𝔾, θ0, NF2@dθ0, NF2@ϕ0, NF2@dϕ0, NF2@𝒜, NF2@ℬ, NF2@𝒞, NF2@𝒟];
 Plot[{(ϕ[T] /. SOLNODES), 𝒞 + (T 𝔾)/2 + (𝒜 Sin[ℬ + T κ])/κ, (θ[T] /. SOLNODES),
   𝒟 + (𝒜 Cos[ℬ + T κ] Sin[θm])/κ}, {T, 0, 20},
  PlotStyle → {Black, Directive[Black, Dashed], Blue, Directive[Blue, Dashed]},
  AxesLabel → {Stl["T"], ""},
  PlotLegends → {"ϕ numerical", "ϕ approximate", "θ numerical", "θ approximate"},
  PlotLabel → lab]]
```



Figure 4 The discrepancies between the numerical and Morin's approximate results are reasonably small.

At the end of *Section 9.7.7 Nutation,* Morin presents a detailed discussion of how his approximate model can be used to illustrate various precession-nutation behaviors of a heavy top. I will content myself here with a Mathematica "Manipulate" that allows users to change the initial conditions of the heavy symmetric top and visualize the corresponding path the tip of the top will take in its body frame.

```
MManipulate[
 ShowPrecessionNutation[θ0, dθ0, 0, dϕ0],
 {{θ0, π / 6., "θ0"}, 0, π, π/32., Appearance → "Labeled"},
 {{dθ0, 0.1, "dθ0"}, -0.2, 0.2, π/32., Appearance → "Labeled"},
 {{dϕ0, 0.05, "dϕ0"}, -0.2, 0.2, π/32., Appearance → "Labeled"}]
```

In[62]:=

```
DynamicModule[{dθ0 = 0.1`, dϕ0 = 0.05`, θ0 = 0.5235987755982988`},
 ShowPrecessionNutation[θ0, dθ0, 0, dϕ0]]
```

Out[62]=



In[57]:=

```
Clear[ShowPrecessionNutation];
ShowPrecessionNutation[θ0_, dθ0_, ϕ0_, dϕ0_] :=
 Module[{θm = π / 6, κ = 8/5, 𝔾 = 0.1, TMax = 20,
   scale = 1.25, δ = 0.1, viewPoint = {2.4, 1, 1}, 𝒦, eqns, soln, O,
   ex, ey, ez, axes, StoC, GBackground, 𝒜, ℬ, 𝒞, 𝒟, θFcn, ϕFcn, lab, G},
  StoC[r_, θ_, ϕ_] := {r Cos[ϕ] Sin[θ], r Sin[θ] Sin[ϕ], r Cos[θ]};

  𝒜 = 1/2 Csc[θm] √(((-2 dϕ0 + 𝔾)^2 + 4 dθ0^2 Csc[θm]^2) Sin[θm]^2);
  𝒞 = ϕ0 + (dθ0 Csc[θm])/κ;
  𝒟 = (2 θ0 κ + (-2 dϕ0 + 𝔾) Sin[θm])/(2 κ);
  ℬ = ArcTan[((2 dϕ0 - 𝔾) Sin[θm]) / (√(4 dθ0^2 + (-2 dϕ0 + 𝔾)^2 Sin[θm]^2)),
    -((2 dθ0) / (√(4 dθ0^2 + (-2 dϕ0 + 𝔾)^2 Sin[θm]^2)))];


  θFcn[T_] := 𝒟 + (𝒜 Cos[ℬ + T κ] Sin[θm])/κ;
  ϕFcn[T_] := 𝒞 + (T 𝔾)/2 + (𝒜 Sin[ℬ + T κ])/κ;

  {O, ex, ey, ez} = {{0, 0, 0}, {1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
  axes = {Black, Line[scale {-ex, ex}], Line[scale {-ey, ey}],
    Line[scale {- ez, ez}], Stl@Text["x", (scale + δ) ex],
    Stl@ Text["y", (scale + δ) ey], Stl@Text["z", (scale + δ) ez]};
  lab = Stl["Precession and nutation of Heavy Top"];
  GBackground = Graphics3D[{{Opacity[.25], Sphere[]}, axes},
    ViewPoint → viewPoint, Boxed → False, PlotLabel → lab];

  G = ParametricPlot3D[
    (StoC[1, θFcn[T], ϕFcn[T]] /. SOLNODES), {T, 0, TMAX}, PlotStyle → Black];
  Show[GBackground, G]]
```

Visualizations

```
Clear[ShowHeavyTop];
ShowHeavyTop[] :=
 Module[{a = 1, b = 0.5, r = 1, scale = 2, δ = 0.1, vp = {2.5, 1.0, 1},
   szArrow = 0.03, O, COM, ex, ey, ez, axesxyz, axes123, θ, range,
   latlongcurves, top, ℛ, θArcFcn, θArc, arcRotation, FVec, base, lab, G},

  range = scale {{- 1, 1}, {- 1, 1}, {- 0.25, 1}};
  {O, ex, ey, ez} = {{0, 0, 0}, {1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
  (* lab frame *)
  axesxyz =
   {Black, Line[scale {-ex, ex}], Line[scale { -ey, ey}], Line[scale {- ez, ez}],
    Tex["x", (scale + δ) ex], Tex["y", (scale + δ) ey], Tex["z", (scale + δ) ez] };

  latlongcurves =
   {LightGray, Table[Line@Table[{a Sin[θθ] Cos[ϕϕ], a Sin[θθ] Sin[ϕϕ], b Cos[θθ] },
       {θθ, 0, π, π/100}], {ϕϕ, 0, 2 π, π/12}],
    Table[Line@Table[{a Sin[θθ] Cos[ϕϕ], a Sin[θθ] Sin[ϕϕ], b Cos[θθ] },
       {ϕϕ, 0, 2 π, π/100}], {θθ, -π, π, π/8}]};
  (* body frame *)
  axes123 = {Black, Line[{-ex, ex}], Line[{ -ey, ey}],
    Line[{- ez, ez}], Tex["e1", 1.1 ex], Tex["e2", 1.1 ey], Tex["e3", 1.1 ez] };

  θ = -π / 6;
  (* In order to generate an arc denoting the θ rotation of the ellipsoid,
  generate a vector that initially points in the ez
   direction and that then is rotated about the ex axis *)
  arcRotation = {Directive[Black, Thin], Arrowheads[Small],
    Arrow@Table[{0.2 Cos[ϕ], 0.2 Sin[ϕ], 0.75}, {ϕ, π / 4, 2 π, 2 π / 64}]};
  θArcFcn[β_] := Dot[RotationMatrix[β, ex], ez];
  θArc = {Tex["θ", 0.8 θArcFcn[β] /. β → θ/2],
    Arrowheads[Small], Arrow@Table[0.7 θArcFcn[β], {β, 0, θ, -0.05}]};
  base = {LightGreen, Polygon[2 {{-1, -1, 0}, {-1, 1, 0}, {1, 1, 0}, {1, -1, 0}}]};
  top = {arcRotation, {Opacity[0.25, Blue], Ellipsoid[O, {a, a, b}]},
    latlongcurves, {Black, axes123}, {Blue, Tube[{{0, 0, r}, {0, 0, -r}}]}};
  top = GeometricTransformation[top, TranslationTransform[{0, 0, r}]];
  top = Rotate[top, θ, ex];
  COM = {0, -r Sin[θ], r Cos[θ]};
  FVec = With[{st = COM, fn = COM + {0, 0, -0.5}, sz = 0.01, szArr = 0.02},
    {Red, Vec[{st, fn}, sz, szArr], Black, Tex["m g⃗", st + 1.1 (fn - st)]}];

  lab = Stl["Oblate Heavy Top"];
  Graphics3D[{axesxyz, base, top, θArc, FVec},
   ImageSize → 500, Axes → False, Boxed → False, PlotRange → range,
   SphericalRegion → True, ViewPoint → vp, PlotLabel → lab]]
```

Graphics Utilities

In[47]:=

```
(* Helper Functionsc*)
Clear[StoC, Tex, Vec, VecLab, Mess, DimMarker3];
StoC[r_, θ_, ϕ_] := {r Sin[θ] Cos[ϕ], r Sin[θ] Sin[ϕ], r Cos[θ] };
Tex[text_, position_] := Text[Style[text, Bold, FontSize → 10], position];
Vec[vec_] := {Arrowheads[0.05], Arrow[Tube[vec, 0.02]]};
Vec[vec_, size_, sizeAH_] := {Arrowheads[sizeAH], Arrow[Tube[vec, size]]};
(* Draw vector with label place beyond the tip *)
VecLab[st_, fn_, sz_, szArrow_, txt_, txtScale_] :=
 Module[{vec, vecLabel},
  vec = Vec[{st, fn}, sz, szArrow];
  vecLabel = Text[Stl[txt], st + txtScale (fn - st)];
  {vec, vecLabel}]
Mess[lab_, Pobj_, offset_] :=
  Module[{Ptex = Pobj + offset, gText, pointer, dirVec, arrow},
   dirVec = Pobj - Ptex;
   arrow = {Arrowheads[Small], Arrow[{Ptex + 0.2 dirVec, Ptex + 0.8 dirVec}]};
   gText = {Black, Style[Text[lab, Ptex], 10, Italic], arrow}];
DimMarker3[{tail_, head_}, lab_, frac_, offset_] :=
 Module[{labPosn},
  labPosn = tail + frac (head - tail);
  {Arrowheads[{-0.01, 0.01}], Arrow[{tail, head}], Text[lab, labPosn + offset]} ]
```